LE EQUAZIONI GONIOMETRICHE CON WIRIS

Con Wiris si possono trovare le soluzioni della maggior parte delle equazioni goniometriche e nel foglio di lavoro che segue puoi vedere alcuni esempi.
Le soluzioni trovate appartengono di solito all'intervallo $[0, 2\pi]$ oppure $[-\pi, \pi]$; in ogni caso vengono trovate le soluzioni principali in un arco di ampiezza 2π.
Osserva che le soluzioni vengono sempre restituite in radianti; per convertire in gradi si usa il comando

convertire $(n,°)$

dove n rappresenta il valore da convertire.



I TRIANGOLI RETTANGOLI CON EXCEL

Con riferimento alla ***figura 1***, i casi che si possono presentare sono i seguenti.

Figura 1



- Conosciamo la misura dell'ipotenusa e quella di un angolo acuto, cioè conosciamo a e β. Ricaviamo che

  $\gamma = 90° - \beta \qquad b = a \sin \beta \qquad c = a \cos \beta$

- Conosciamo la misura dell'ipotenusa e quella di un cateto, cioè conosciamo a e b. Ricaviamo che

  $c = \sqrt{a^2 - b^2} \qquad \sin \beta = \dfrac{b}{a} \qquad \gamma = 90° - \beta$

- Conosciamo la misura dei cateti, cioè conosciamo b e c. Ricaviamo che

  $a = \sqrt{c^2 + b^2} \qquad \tan \beta = \dfrac{b}{c} \qquad \gamma = 90° - \beta$

- Conosciamo la misura di un cateto e quella di un angolo acuto, cioè conosciamo b e γ. Ricaviamo che

  $\beta = 90° - \gamma \qquad c = b \tan \gamma \qquad a = \dfrac{b}{\cos \gamma}$

Apriamo allora un foglio di lavoro e impostiamo il calcolo inserendo stringhe, dati e formule nelle celle specificate, come è indicato di seguito.
Nella preparazione del foglio, di cui puoi vedere un esempio in figura, abbiamo tenuto conto del fatto che le funzioni goniometriche di **Excel** usano angoli la cui misura è espressa in radianti mentre noi prevediamo di assegnare le misure

degli angoli in gradi (abbreviato nel foglio di esempio in "gr"); quando uno dei valori noti è un angolo, è quindi prevista una cella in cui calcolare il corrispondente valore dell'angolo in radianti (abbreviato nel foglio di esempio in "rad").

	A	B	C	D	E	F	G	H
1			**RISOLUZIONE TRIANGOLI RETTANGOLI**					
2								
3	**1° CASO - ipotenusa e angolo acuto: a, beta**					**RISULTATI**		
4	a	beta (gr)	beta (rad)			gamma (gr)	b	c
5	10	27,5	0,4799655			62,5	4,6174861	8,870108
6								
7	**2° CASO - ipotenusa e cateto: a, b**					**RISULTATI**		
8	a	b				beta (gr)	gamma (gr)	c
9	15	12				53,130102	36,869898	9
10								
11	**3° CASO - i due cateti: b, c**					**RISULTATI**		
12	b	c				beta (gr)	gamma (gr)	a
13	7	9				37, 874984	52,125016	11,40175
14								
15	**4° CASO - cateto e angolo acuto: b, gamma**					**RISULTATI**		
16	b	gamma (gr)	gamma (rad)			beta (gr)	c	a
17	18	36,57	0,6382669			53,43	13,353362	22,41232
18								

La funzione di Excel che esegue la conversione da gradi a radianti è la funzione **RADIANTI**(angolo), quella che esegue la conversione da radianti a gradi è la funzione **GRADI**(angolo).
Le funzioni di Excel che consentono di ricavare l'ampiezza di un angolo nota una delle sue funzioni goniometriche sono:

- ARCSEN(x)
- ARCCOS(x)
- ARCTAN(x)

dove x è il valore della funzione goniometrica. Per esempio ARCSEN(1/2) restituisce l'angolo il cui seno vale $\frac{1}{2}$.

Relativamente al primo caso, abbiamo posto in A5 la misura dell'ipotenusa a (10) e in B5 la misura in gradi nella forma decimale dell'angolo β (27,5). Le formule da inserire sono poi le seguenti:

C5	= RADIANTI(B5)	(formula per trasformare la misura di β in radianti)
F5	= 90 − B5	(formula per il calcolo di γ in gradi)
G5	= $A5 * $SEN (C5)	(formula per il calcolo di b)
H5	= $A5 * $COS (C5)	(formula per il calcolo di c)

Prosegui impostando gli altri casi come è illustrato nell'esempio; ti indichiamo solamente le formule da inserire nelle celle specificate lasciando a te il compito di inserire le stringhe.

F9	= GRADI(ARCSEN(B9/A9))	(calcolo di β in gradi)
G9	= 90 − F9	(calcolo di γ in gradi)
H9	= RADQ(A9 * A9 − B9 * B9)	(calcolo di c)
F13	= GRADI(ARCTAN(A13/B13))	(calcolo di β in gradi)
G13	= 90 − F13	(calcolo di γ in gradi)
H13	= RADQ(A13 * A13 + B13 * B13)	(calcolo di a)
C17	= RADIANTI(B17)	(conversione in radianti della misura di γ)
F17	= 90 − B17	(calcolo di β in gradi)
G17	= A17 * TAN(C17)	(calcolo di c)
H17	= A17/COS(C17)	(calcolo di a)

I TRIANGOLI QUALUNQUE CON EXCEL

Con riferimento alla ***figura 2***, i casi che si possono presentare nella risoluzione di un triangolo qualsiasi sono i seguenti.

Figura 2



- Conosciamo la misura di due angoli e quella di un lato, ad esempio α, γ e b. Ricaviamo che

$$\beta = 180° - (\alpha + \gamma) \qquad a = \frac{b \sin \alpha}{\sin \beta} \qquad c = \frac{b \sin \gamma}{\sin \beta}$$

- Conosciamo la misura di due lati e quella dell'angolo compreso, ad esempio a, c e β. Usando il teorema di Carnot:

$$b = \sqrt{a^2 + c^2 - 2ac \cos \beta} \qquad \cos \alpha = \frac{b^2 + c^2 - a^2}{2bc} \qquad \gamma = 180° - (\alpha + \beta)$$

- Conosciamo la misura dei tre lati, cioè conosciamo a, b e c. Usando il teorema di Carnot ricaviamo che:

$$\cos \alpha = \frac{b^2 + c^2 - a^2}{2bc} \qquad \cos \beta = \frac{a^2 + c^2 - b^2}{2ac} \qquad \gamma = 180° - (\alpha + \beta)$$

- Conosciamo la misura di due lati e dell'angolo opposto ad uno di essi, ad esempio a, b e α. Ricaviamo che:
  - se $a \geq b$ esiste un solo triangolo (***figura 3a.***) ed è:

$$\sin \beta = \frac{b \sin \alpha}{a} \quad \text{cioè} \quad \beta = \arcsin\left(\frac{b \sin \alpha}{a}\right); \quad \gamma = 180° - (\alpha + \beta) \qquad c = \frac{a \sin \gamma}{\sin \alpha}$$

Figura 3



  - se $b \sin \alpha \leq a < b$ esistono due triangoli (***figura 3b.*** e ***c.***)
    - il primo si risolve con le stesse modalità del caso precedente
    - per il secondo $\quad \beta' = 180° - \beta \qquad \gamma' = 180° - (\alpha + \beta') \qquad c = \dfrac{a \sin \gamma'}{\sin \alpha}$
  - se $a < b \sin \alpha$ il problema non ammette soluzione (***figura 3d.***).

Impostiamo il foglio di lavoro in questo modo (osserva la figura per inserire le stringhe, i dati e convertire gli angoli, noi ti indichiamo solamente le formule di calcolo degli elementi del triangolo)

1° CASO

G5 = 180 – B5 – C5
H5 = A5 * SEN(D5)/SEN(RADIANTI(G5))
I5 = A5 * SEN(E5)/SEN(RADIANTI(G5))

2° CASO

G9 = RADQ(A9 * A9 + B9 * B9 – 2 * A9 * B9 * COS(D9))
H9 = GRADI(ARCCOS((G9 * G9 + B9 * B9 – A9 * A9)/(2 * G9 * B9)))
I9 = 180 – C9 – H9

3° CASO

G13 $= \text{GRADI}(\text{ARCCOS}((B13 * B13 + C13 * C13 - A13 * A13)/(2 * B13 * C13)))$
H13 $= \text{GRADI}(\text{ARCCOS}((A13 * A13 + C13 * C13 - B13 * B13)/(2 * A13 * C13)))$
I13 $= 180 - G13 - H13$

4° CASO

Questo è il caso più complesso perché, a seconda delle misure assegnate, dobbiamo prevedere di risolvere un solo triangolo, due triangoli o nessun triangolo.

Formule per il caso di un solo triangolo (caso $a \geq b$):

G17 $= \text{SE}(A17 >= B17; \text{GRADI}(\text{ARCSEN}((B17 * \text{SEN}(D17))/A17)))$
H17 $= \text{SE}(A17 >= B17; 180 - G17 - C17)$
I17 $= \text{SE}(A17 >= B17; (A17 * \text{SEN}(\text{RADIANTI}(H17)))/\text{SEN}(D17))$

Formule per il caso di due triangoli (caso $b \sin\alpha \leq a < b$):

G19 $= \text{SE}(\text{E}(A17 < B17; A17 >= B17 * \text{SEN}(D17)); \text{GRADI}(\text{ARCSEN}((B17 * \text{SEN}(D17))/A17)))$
H19 $= \text{SE}(\text{E}(A17 < B17; A17 >= B17 * \text{SEN}(D17)); 180 - G19 - C17)$
I19 $= \text{SE}(\text{E}(A17 < B17; A17 >= B17 * \text{SEN}(D17)); (A17 * \text{SEN}(\text{RADIANTI}(H19)))/\text{SEN}(D17))$
G20 $= \text{SE}(G19; 180 - G19)$
H20 $= \text{SE}(G19; 180 - C17 - G20)$
I20 $= \text{SE}(G19; A17 * \text{SEN}(\text{RADIANTI}(H20))/\text{SEN}(D17))$

Formule per il caso di nessun triangolo:

G22 $= \text{SE}(A17 < B17 * \text{SEN}(D17); \text{VERO}())$

	A	B	C	D	E	F	G	H	I
1			**RISOLUZIONE TRIANGOLI QUALSIASI**						
2									
3	**1° CASO - due angoli e un lato: alfa, gamma, b**							**RISULTATI**	
4	b	alfa (gr)	gamma (gr)	alfa (rad)	gamma (rad)		beta (gr)	a	c
5	15	32,25	50,65	0,562869	0,8840093		97,1	8,066069	11,68894
6									
7	**2° CASO - due lati e l'angolo compreso: a, c, beta**							**RISULTATI**	
8	a	c	beta (gr)	beta (rad)			b	alfa (gr)	gamma (gr)
9	153	75	15,22056	0,265649			83,00019	151,0563	13,72313
10									
11	**3° CASO - tre lati: a, b, c**							**RISULTATI**	
12	a	b	c				alfa (gr)	beta (gr)	gamma (gr)
13	175	286	197				37,01136	100,3284	42,66027
14									
15	**4° CASO - due lati e l'angolo opposto ad uno di essi: a, b, alfa**							**RISULTATI**	
16	a	b	alfa (gr)	alfa (rad)			beta (gr)	gamma (gr)	c
17	98	112	55	0,959931		1 triangolo	FALSO	FALSO	FALSO
18									
19						2 triangoli	69,41861	55,58139	98,69125
20							110,5814	14,41861	29,78988
21									
22						nes. triang.	FALSO		